

regress postestimation time series — Postestimation tools for regress with time series

- | | | |
|--|---|--|
| Postestimation commands estat durbinalt Stored results References | estat archlm estat dwatson Methods and formulas Also see | estat bgodfrey Remarks and examples Acknowledgment |
|--|---|--|

Postestimation commands

The following postestimation commands for time series are available for `regress`:

| Command | Description |
|------------------------------|---|
| <code>estat archlm</code> | test for ARCH effects in the residuals |
| <code>estat bgodfrey</code> | Breusch–Godfrey test for higher-order serial correlation |
| <code>estat durbinalt</code> | Durbin’s alternative test for serial correlation |
| <code>estat dwatson</code> | Durbin–Watson <i>d</i> statistic to test for first-order serial correlation |
| <code>estat sbknown</code> | perform tests for a structural break with a known break date |
| <code>estat single</code> | perform tests for a structural break with an unknown break date |

These commands provide regression diagnostic tools specific to time series. You must `tsset` your data before using these commands; see [TS] [tsset](#).

estat archlm

Description for estat archlm

`estat archlm` performs Engle’s Lagrange multiplier (LM) test for the presence of autoregressive conditional heteroskedasticity.

Menu for estat

Statistics > Postestimation

Syntax for estat archlm

```
estat archlm [ , archlm_options ]
```

| <i>archlm_options</i> | Description |
|-----------------------------------|--|
| <code>lags(<i>numlist</i>)</code> | test <i>numlist</i> lag orders |
| <code>force</code> | allow test after <code>regress</code> , <code>vce(robust)</code> |

Options for estat archlm

`lags(numlist)` specifies a list of numbers, indicating the lag orders to be tested. The test will be performed separately for each order. The default is order one.

`force` allows the test to be run after `regress, vce(robust)`. The command will not work if the `vce(cluster clustvar)` option is specified with `regress`; see [R] [regress](#).

estat bgodfrey

Description for estat bgodfrey

`estat bgodfrey` performs the Breusch–Godfrey test for higher-order serial correlation in the disturbance. This test does not require that all the regressors be strictly exogenous.

Menu for estat

Statistics > Postestimation

Syntax for estat bgodfrey

```
estat bgodfrey [, bgodfrey_options]
```

| <i>bgodfrey_options</i> | Description |
|--------------------------------|---|
| <u>lags</u> (<i>numlist</i>) | test <i>numlist</i> lag orders |
| <u>nomiss0</u> | do not use Davidson and MacKinnon's approach |
| <u>small</u> | obtain <i>p</i> -values using the <i>F</i> or <i>t</i> distribution |

Options for estat bgodfrey

`lags(numlist)` specifies a list of numbers, indicating the lag orders to be tested. The test will be performed separately for each order. The default is order one.

`nomiss0` specifies that Davidson and MacKinnon's approach (1993, 358), which replaces the missing values in the initial observations on the lagged residuals in the auxiliary regression with zeros, not be used.

`small` specifies that the *p*-values of the test statistics be obtained using the *F* or *t* distribution instead of the default chi-squared or normal distribution.

estat durbinalt

Description for estat durbinalt

`estat durbinalt` performs Durbin's alternative test for serial correlation in the disturbance. This test does not require that all the regressors be strictly exogenous.

Menu for estat

Statistics > Postestimation

Syntax for estat durbinalt

```
estat durbinalt [ , durbinalt_options ]
```

| <i>durbinalt_options</i> | Description |
|-----------------------------------|--|
| <code>lags(<i>numlist</i>)</code> | test <i>numlist</i> lag orders |
| <code>nomiss0</code> | do not use Davidson and MacKinnon's approach |
| <code>robust</code> | compute standard errors using the robust/sandwich estimator |
| <code>small</code> | obtain <i>p</i> -values using the <i>F</i> or <i>t</i> distribution |
| <code>force</code> | allow test after <code>regress</code> , <code>vce(robust)</code> or after <code>newey</code> |

Options for estat durbinalt

`lags(numlist)` specifies a list of numbers, indicating the lag orders to be tested. The test will be performed separately for each order. The default is order one.

`nomiss0` specifies that Davidson and MacKinnon's approach (1993, 358), which replaces the missing values in the initial observations on the lagged residuals in the auxiliary regression with zeros, not be used.

`robust` specifies that the Huber/White/sandwich robust estimator for the variance–covariance matrix be used in Durbin's alternative test.

`small` specifies that the *p*-values of the test statistics be obtained using the *F* or *t* distribution instead of the default chi-squared or normal distribution. This option may not be specified with `robust`, which always uses an *F* or a *t* distribution.

`force` allows the test to be run after `regress`, `vce(robust)` and after `newey` (see [R] `regress` and [TS] `newey`). The command will not work if the `vce(cluster clustvar)` option is specified with `regress`.

estat dwatson

Description for estat dwatson

`estat dwatson` computes the Durbin–Watson d statistic (Durbin and Watson 1950) to test for first-order serial correlation in the disturbance when all the regressors are strictly exogenous.

Menu for estat

Statistics > Postestimation

Syntax for estat dwatson

```
estat dwatson
```

Remarks and examples

stata.com

The Durbin–Watson test is used to determine whether the error term in a linear regression model follows an AR(1) process. For the linear model

$$y_t = \mathbf{x}_t\boldsymbol{\beta} + u_t$$

the AR(1) process can be written as

$$u_t = \rho u_{t-1} + \epsilon_t$$

In general, an AR(1) process requires only that ϵ_t be independent and identically distributed (i.i.d.). The Durbin–Watson test, however, requires ϵ_t to be distributed $N(0, \sigma^2)$ for the statistic to have an exact distribution. Also, the Durbin–Watson test can be applied only when the regressors are strictly exogenous. A regressor x is strictly exogenous if $\text{Corr}(x_s, u_t) = 0$ for all s and t , which precludes the use of the Durbin–Watson statistic with models where lagged values of the dependent variable are included as regressors.

The null hypothesis of the test is that there is no first-order autocorrelation. The Durbin–Watson d statistic can take on values between 0 and 4 and under the null d is equal to 2. Values of d less than 2 suggest positive autocorrelation ($\rho > 0$), whereas values of d greater than 2 suggest negative autocorrelation ($\rho < 0$). Calculating the exact distribution of the d statistic is difficult, but empirical upper and lower bounds have been established based on the sample size and the number of regressors. Extended tables for the d statistic have been published by Savin and White (1977). For example, suppose you have a model with 30 observations and three regressors (including the constant term). For a test of the null hypothesis of no autocorrelation versus the alternative of positive autocorrelation, the lower bound of the d statistic is 1.284, and the upper bound is 1.567 at the 5% significance level. You would reject the null if $d < 1.284$, and you would fail to reject if $d > 1.567$. A value falling within the range (1.284, 1.567) leads to no conclusion about whether or not to reject the null hypothesis.

When lagged dependent variables are included among the regressors, the past values of the error term are correlated with those lagged variables at time t , implying that they are not strictly exogenous regressors. The inclusion of covariates that are not strictly exogenous causes the d statistic to be biased toward the acceptance of the null hypothesis. Durbin (1970) suggested an alternative test for models with lagged dependent variables and extended that test to the more general AR(p) serial correlation process

$$u_t = \rho_1 u_{t-1} + \cdots + \rho_p u_{t-p} + \epsilon_t$$

where ϵ_t is i.i.d. with variance σ^2 but is not assumed or required to be normal for the test.

The null hypothesis of Durbin’s alternative test is

$$H_0 : \rho_1 = 0, \dots, \rho_p = 0$$

and the alternative is that at least one of the ρ ’s is nonzero. Although the null hypothesis was originally derived for an AR(p) process, this test turns out to have power against MA(p) processes as well. Hence, the actual null of this test is that there is no serial correlation up to order p because the MA(p) and the AR(p) models are locally equivalent alternatives under the null. See [Godfrey \(1988, 113–115\)](#) for a discussion of this result.

Durbin’s alternative test is in fact a LM test, but it is most easily computed with a Wald test on the coefficients of the lagged residuals in an auxiliary OLS regression of the residuals on their lags and all the covariates in the original regression. Consider the linear regression model

$$y_t = \beta_1 x_{1t} + \dots + \beta_k x_{kt} + u_t \quad (1)$$

in which the covariates x_1 through x_k are not assumed to be strictly exogenous and u_t is assumed to be i.i.d. and to have finite variance. The process is also assumed to be stationary. (See [Wooldridge \[2016, sec. 11.1\]](#) for a discussion of stationarity.) Estimating the parameters in (1) by OLS obtains the residuals \hat{u}_t . Next another OLS regression is performed of \hat{u}_t on $\hat{u}_{t-1}, \dots, \hat{u}_{t-p}$ and the other regressors,

$$\hat{u}_t = \gamma_1 \hat{u}_{t-1} + \dots + \gamma_p \hat{u}_{t-p} + \beta_1 x_{1t} + \dots + \beta_k x_{kt} + \epsilon_t \quad (2)$$

where ϵ_t stands for the random-error term in this auxiliary OLS regression. Durbin’s alternative test is then obtained by performing a Wald test that $\gamma_1, \dots, \gamma_p$ are jointly zero. The test can be made robust to an unknown form of heteroskedasticity by using a robust VCE estimator when estimating the regression in (2). When there are only strictly exogenous regressors and $p = 1$, this test is asymptotically equivalent to the Durbin–Watson test.

The Breusch–Godfrey test is also an LM test of the null hypothesis of no autocorrelation versus the alternative that u_t follows an AR(p) or MA(p) process. Like Durbin’s alternative test, it is based on the auxiliary regression (2), and it is computed as NR^2 , where N is the number of observations and R^2 is the simple R^2 from the regression. This test and Durbin’s alternative test are asymptotically equivalent. The test statistic NR^2 has an asymptotic χ^2 distribution with p degrees of freedom. It is valid with or without the strict exogeneity assumption but is not robust to conditional heteroskedasticity, even if a robust VCE is used when fitting (2).

In fitting (2), the values of the lagged residuals will be missing in the initial periods. As noted by [Davidson and MacKinnon \(1993\)](#), the residuals will not be orthogonal to the other covariates in the model in this restricted sample, which implies that the R^2 from the auxiliary regression will not be zero when the lagged residuals are left out. Hence, Breusch and Godfrey’s NR^2 version of the test may overreject in small samples. To correct this problem, [Davidson and MacKinnon \(1993\)](#) recommend setting the missing values of the lagged residuals to zero and running the auxiliary regression in (2) over the full sample used in (1). This small-sample correction has become conventional for both the Breusch–Godfrey and Durbin’s alternative test, and it is the default for both commands. Specifying the `nomiss0` option overrides this default behavior and treats the initial missing values generated by regressing on the lagged residuals as missing. Hence, `nomiss0` causes these initial observations to be dropped from the sample of the auxiliary regression.

Durbin’s alternative test and the Breusch–Godfrey test were originally derived for the case covered by `regress` without the `vce(robust)` option. However, after `regress`, `vce(robust)` and `newey`, Durbin’s alternative test is still valid and can be invoked if the `robust` and `force` options are specified.

▷ Example 1: tests for serial correlation

Using data from [Klein \(1950\)](#), we first fit an OLS regression of consumption on the government wage bill:

```
. use http://www.stata-press.com/data/r14/klein
. tsset yr
      time variable:  yr, 1920 to 1941
      delta: 1 unit
. regress consump wagegovt
```

| Source | SS | df | MS | Number of obs | = | 22 |
|----------|------------|----|------------|---------------|---|--------|
| Model | 532.567711 | 1 | 532.567711 | F(1, 20) | = | 17.72 |
| Residual | 601.207167 | 20 | 30.0603584 | Prob > F | = | 0.0004 |
| | | | | R-squared | = | 0.4697 |
| | | | | Adj R-squared | = | 0.4432 |
| Total | 1133.77488 | 21 | 53.9892799 | Root MSE | = | 5.4827 |

| consump | Coef. | Std. Err. | t | P> t | [95% Conf. Interval] |
|----------|----------|-----------|-------|-------|----------------------|
| wagegovt | 2.50744 | .5957173 | 4.21 | 0.000 | 1.264796 3.750085 |
| _cons | 40.84699 | 3.192183 | 12.80 | 0.000 | 34.18821 47.50577 |

If we assume that `wagegovt` is a strictly exogenous variable, we can use the Durbin–Watson test to check for first-order serial correlation in the errors.

```
. estat dwatson
Durbin-Watson d-statistic( 2, 22) = .3217998
```

The Durbin–Watson d statistic, 0.32, is far from the center of its distribution ($d = 2.0$). Given 22 observations and two regressors (including the constant term) in the model, the lower 5% bound is about 0.997, much greater than the computed d statistic. Assuming that `wagegovt` is strictly exogenous, we can reject the null of no first-order serial correlation. Rejecting the null hypothesis does not necessarily mean an AR process; other forms of misspecification may also lead to a significant test statistic. If we are willing to assume that the errors follow an AR(1) process and that `wagegovt` is strictly exogenous, we could refit the model using `arima` or `prais` and model the error process explicitly; see [\[TS\] arima](#) and [\[TS\] prais](#).

If we are not willing to assume that `wagegovt` is strictly exogenous, we could instead use Durbin’s alternative test or the Breusch–Godfrey to test for first-order serial correlation. Because we have only 22 observations, we will use the `small` option.

```
. estat durbinalt, small
Durbin’s alternative test for autocorrelation
```

| lags(p) | F | df | Prob > F |
|---------|--------|-----------|----------|
| 1 | 35.035 | (1, 19) | 0.0000 |

H0: no serial correlation

```
. estat bgodfrey, small
Breusch-Godfrey LM test for autocorrelation
```

| lags(p) | F | df | Prob > F |
|---------|--------|-----------|----------|
| 1 | 14.264 | (1, 19) | 0.0013 |

H0: no serial correlation

Both tests strongly reject the null of no first-order serial correlation, so we decide to refit the model with two lags of `consump` included as regressors and then rerun `estat durbinalt` and `estat bgodfrey`. Because the revised model includes lagged values of the dependent variable, the Durbin–Watson test is not applicable.

```
. regress consump wagegovt L.consump L2.consump
```

| Source | SS | df | MS | Number of obs | = | 20 |
|----------|------------|----|------------|---------------|---|--------|
| Model | 702.660311 | 3 | 234.220104 | F(3, 16) | = | 44.01 |
| Residual | 85.1596011 | 16 | 5.32247507 | Prob > F | = | 0.0000 |
| | | | | R-squared | = | 0.8919 |
| | | | | Adj R-squared | = | 0.8716 |
| Total | 787.819912 | 19 | 41.4642059 | Root MSE | = | 2.307 |

| consump | Coef. | Std. Err. | t | P> t | [95% Conf. Interval] |
|----------|----------|-----------|-------|-------|----------------------|
| wagegovt | .6904282 | .3295485 | 2.10 | 0.052 | -.0081835 1.38904 |
| consump | | | | | |
| L1. | 1.420536 | .197024 | 7.21 | 0.000 | 1.002864 1.838208 |
| L2. | -.650888 | .1933351 | -3.37 | 0.004 | -1.06074 -.241036 |
| _cons | 9.209073 | 5.006701 | 1.84 | 0.084 | -1.404659 19.82281 |

```
. estat durbinalt, small lags(1/2)
Durbin's alternative test for autocorrelation
```

| lags(p) | F | df | Prob > F |
|---------|-------|-----------|----------|
| 1 | 0.080 | (1, 15) | 0.7805 |
| 2 | 0.260 | (2, 14) | 0.7750 |

H0: no serial correlation

```
. estat bgodfrey, small lags(1/2)
Breusch-Godfrey LM test for autocorrelation
```

| lags(p) | F | df | Prob > F |
|---------|-------|-----------|----------|
| 1 | 0.107 | (1, 15) | 0.7484 |
| 2 | 0.358 | (2, 14) | 0.7056 |

H0: no serial correlation

Although `wagegovt` and the constant term are no longer statistically different from zero at the 5% level, the output from `estat durbinalt` and `estat bgodfrey` indicates that including the two lags of `consump` has removed any serial correlation from the errors.

Engle (1982) suggests an LM test for checking for autoregressive conditional heteroskedasticity (ARCH) in the errors. The p th-order ARCH model can be written as

$$\begin{aligned}\sigma_t^2 &= E(u_t^2 | u_{t-1}, \dots, u_{t-p}) \\ &= \gamma_0 + \gamma_1 u_{t-1}^2 + \dots + \gamma_p u_{t-p}^2\end{aligned}$$

To test the null hypothesis of no autoregressive conditional heteroskedasticity (that is, $\gamma_1 = \dots = \gamma_p = 0$), we first fit the OLS model (1), obtain the residuals \hat{u}_t , and run another OLS regression on the lagged residuals:

$$\hat{u}_t^2 = \gamma_0 + \gamma_1 \hat{u}_{t-1}^2 + \dots + \gamma_p \hat{u}_{t-p}^2 + \epsilon \quad (3)$$

The test statistic is NR^2 , where N is the number of observations in the sample and R^2 is the R^2 from the regression in (3). Under the null hypothesis, the test statistic follows a χ_p^2 distribution.

► Example 2: estat archlm

We refit the original model that does not include the two lags of `consump` and then use `estat archlm` to see if there is any evidence that the errors are autoregressive conditional heteroskedastic.

```
. regress consump wagegovt
```

| Source | SS | df | MS | Number of obs | = | 22 |
|----------|------------|----|------------|---------------|---|--------|
| Model | 532.567711 | 1 | 532.567711 | F(1, 20) | = | 17.72 |
| Residual | 601.207167 | 20 | 30.0603584 | Prob > F | = | 0.0004 |
| | | | | R-squared | = | 0.4697 |
| | | | | Adj R-squared | = | 0.4432 |
| Total | 1133.77488 | 21 | 53.9892799 | Root MSE | = | 5.4827 |

| consump | Coef. | Std. Err. | t | P> t | [95% Conf. Interval] |
|----------|----------|-----------|-------|-------|----------------------|
| wagegovt | 2.50744 | .5957173 | 4.21 | 0.000 | 1.264796 3.750085 |
| _cons | 40.84699 | 3.192183 | 12.80 | 0.000 | 34.18821 47.50577 |

```
. estat archlm, lags(1 2 3)
```

LM test for autoregressive conditional heteroskedasticity (ARCH)

| lags(p) | chi2 | df | Prob > chi2 |
|---------|-------|----|-------------|
| 1 | 5.543 | 1 | 0.0186 |
| 2 | 9.431 | 2 | 0.0090 |
| 3 | 9.039 | 3 | 0.0288 |

H0: no ARCH effects vs. H1: ARCH(p) disturbance

`estat archlm` shows the results for tests of ARCH(1), ARCH(2), and ARCH(3) effects, respectively. At the 5% significance level, all three tests reject the null hypothesis that the errors are not autoregressive conditional heteroskedastic. See [TS] [arch](#) for information on fitting ARCH models.

Stored results

`estat archlm` stores the following in `r()`:

Scalars

| | |
|------------------------|------------------------|
| <code>r(N)</code> | number of observations |
| <code>r(k)</code> | number of regressors |
| <code>r(N_gaps)</code> | number of gaps |

Macros

| | |
|----------------------|-----------|
| <code>r(lags)</code> | lag order |
|----------------------|-----------|

Matrices

| | |
|----------------------|-----------------------------------|
| <code>r(arch)</code> | test statistic for each lag order |
| <code>r(df)</code> | degrees of freedom |
| <code>r(p)</code> | two-sided p -values |

`estat bgodfrey` stores the following in `r()`:

Scalars

| | |
|------------------------|------------------------|
| <code>r(N)</code> | number of observations |
| <code>r(k)</code> | number of regressors |
| <code>r(N_gaps)</code> | number of gaps |

Macros

| | |
|----------------------|-----------|
| <code>r(lags)</code> | lag order |
|----------------------|-----------|

Matrices

| | |
|----------------------|---|
| <code>r(chi2)</code> | χ^2 statistic for each lag order |
| <code>r(F)</code> | F statistic for each lag order (small only) |
| <code>r(df_r)</code> | residual degrees of freedom (small only) |
| <code>r(df)</code> | degrees of freedom |
| <code>r(p)</code> | two-sided p -values |

`estat durbinalt` stores the following in `r()`:

Scalars

| | |
|------------------------|------------------------|
| <code>r(N)</code> | number of observations |
| <code>r(k)</code> | number of regressors |
| <code>r(N_gaps)</code> | number of gaps |

Macros

| | |
|----------------------|-----------|
| <code>r(lags)</code> | lag order |
|----------------------|-----------|

Matrices

| | |
|----------------------|---|
| <code>r(chi2)</code> | χ^2 statistic for each lag order |
| <code>r(F)</code> | F statistic for each lag order (small only) |
| <code>r(df_r)</code> | residual degrees of freedom (small only) |
| <code>r(df)</code> | degrees of freedom |
| <code>r(p)</code> | two-sided p -values |

`estat dwatson` stores the following in `r()`:

Scalars

| | |
|------------------------|-------------------------|
| <code>r(N)</code> | number of observations |
| <code>r(k)</code> | number of regressors |
| <code>r(N_gaps)</code> | number of gaps |
| <code>r(dw)</code> | Durbin–Watson statistic |

Methods and formulas

Consider the regression

$$y_t = \beta_1 x_{1t} + \cdots + \beta_k x_{kt} + u_t \quad (4)$$

in which some of the covariates are not strictly exogenous. In particular, some of the x_{it} may be lags of the dependent variable. We are interested in whether the u_t are serially correlated.

The Durbin–Watson d statistic reported by `estat dwatson` is

$$d = \frac{\sum_{t=1}^{n-1} (\hat{u}_{t+1} - \hat{u}_t)^2}{\sum_{t=1}^n \hat{u}_t^2}$$

where \hat{u}_t represents the residual of the t th observation.

To compute Durbin’s alternative test and the Breusch–Godfrey test against the null hypothesis that there is no p th order serial correlation, we fit the regression in (4), compute the residuals, and then fit the following auxiliary regression of the residuals \hat{u}_t on p lags of \hat{u}_t and on all the covariates in the original regression in (4):

$$\hat{u}_t = \gamma_1 \hat{u}_{t-1} + \cdots + \gamma_p \hat{u}_{t-p} + \beta_1 x_{1t} + \cdots + \beta_k x_{kt} + \epsilon \quad (5)$$

Durbin’s alternative test is computed by performing a Wald test to determine whether the coefficients of $\hat{u}_{t-1}, \dots, \hat{u}_{t-p}$ are jointly different from zero. By default, the statistic is assumed to be distributed $\chi^2(p)$. When `small` is specified, the statistic is assumed to follow an $F(p, N - p - k)$ distribution. The reported p -value is a two-sided p -value. When `robust` is specified, the Wald test is performed using the Huber/White/sandwich estimator of the variance–covariance matrix, and the test is robust to an unspecified form of heteroskedasticity.

The Breusch–Godfrey test is computed as NR^2 , where N is the number of observations in the auxiliary regression (5) and R^2 is the R^2 from the same regression (5). Like Durbin’s alternative test, the Breusch–Godfrey test is asymptotically distributed $\chi^2(p)$, but specifying `small` causes the p -value to be computed using an $F(p, N - p - k)$.

By default, the initial missing values of the lagged residuals are replaced with zeros, and the auxiliary regression is run over the full sample used in the original regression of (4). Specifying the `nomiss0` option causes these missing values to be treated as missing values, and the observations are dropped from the sample.

Engle’s LM test for ARCH(p) effects fits an OLS regression of \hat{u}_t^2 on $\hat{u}_{t-1}^2, \dots, \hat{u}_{t-p}^2$:

$$\hat{u}_t^2 = \gamma_0 + \gamma_1 \hat{u}_{t-1}^2 + \cdots + \gamma_p \hat{u}_{t-p}^2 + \epsilon$$

The test statistic is nR^2 and is asymptotically distributed $\chi^2(p)$.

Acknowledgment

The original versions of `estat archlm`, `estat bgodfrey`, and `estat durbinalt` were written by Christopher F. Baum of the Department of Economics at Boston College and author of the Stata Press books *An Introduction to Modern Econometrics Using Stata* and *An Introduction to Stata Programming*.

References

- Baum, C. F. 2006. *An Introduction to Modern Econometrics Using Stata*. College Station, TX: Stata Press.
- Baum, C. F., and V. L. Wiggins. 2000a. sg135: Test for autoregressive conditional heteroskedasticity in regression error distribution. *Stata Technical Bulletin* 55: 13–14. Reprinted in *Stata Technical Bulletin Reprints*, vol. 10, pp. 143–144. College Station, TX: Stata Press.
- . 2000b. sg136: Tests for serial correlation in regression error distribution. *Stata Technical Bulletin* 55: 14–15. Reprinted in *Stata Technical Bulletin Reprints*, vol. 10, pp. 145–147. College Station, TX: Stata Press.
- Beran, R. J., and N. I. Fisher. 1998. A conversation with Geoff Watson. *Statistical Science* 13: 75–93.
- Breusch, T. S. 1978. Testing for autocorrelation in dynamic linear models. *Australian Economic Papers* 17: 334–355.
- Davidson, R., and J. G. MacKinnon. 1993. *Estimation and Inference in Econometrics*. New York: Oxford University Press.
- Durbin, J. 1970. Testing for serial correlation in least-squares regressions when some of the regressors are lagged dependent variables. *Econometrica* 38: 410–421.
- Durbin, J., and S. J. Koopman. 2012. *Time Series Analysis by State Space Methods*. 2nd ed. Oxford: Oxford University Press.
- Durbin, J., and G. S. Watson. 1950. Testing for serial correlation in least squares regression. I. *Biometrika* 37: 409–428.
- . 1951. Testing for serial correlation in least squares regression. II. *Biometrika* 38: 159–177.
- Engle, R. F. 1982. Autoregressive conditional heteroscedasticity with estimates of the variance of United Kingdom inflation. *Econometrica* 50: 987–1007.
- Fisher, N. I., and P. Hall. 1998. Geoffrey Stuart Watson: Tributes and obituary (3 December 1921–3 January 1998). *Australian and New Zealand Journal of Statistics* 40: 257–267.
- Godfrey, L. G. 1978. Testing against general autoregressive and moving average error models when the regressors include lagged dependent variables. *Econometrica* 46: 1293–1301.
- . 1988. *Misspecification Tests in Econometrics: The Lagrange Multiplier Principle and Other Approaches*. Econometric Society Monographs, No. 16. Cambridge: Cambridge University Press.
- Greene, W. H. 2012. *Econometric Analysis*. 7th ed. Upper Saddle River, NJ: Prentice Hall.
- Klein, L. R. 1950. *Economic Fluctuations in the United States 1921–1941*. New York: Wiley.
- Koopman, S. J. 2012. James Durbin, FBA, 1923–2012. *Journal of the Royal Statistical Society, Series A* 175: 1060–1064.
- Phillips, P. C. B. 1988. The ET Interview: Professor James Durbin. *Econometric Theory* 4: 125–157.
- Savin, N. E., and K. J. White. 1977. The Durbin–Watson test for serial correlation with extreme sample sizes or many regressors. *Econometrica* 45: 1989–1996.
- Wooldridge, J. M. 2016. *Introductory Econometrics: A Modern Approach*. 6th ed. Boston: Cengage.

James Durbin (1923–2012) was a British statistician who was born in Wigan, near Manchester. He studied mathematics at Cambridge and after military service and various research posts joined the London School of Economics in 1950. Later in life, he was also affiliated with University College London. His many contributions to statistics centered on serial correlation, time series (including major contributions to structural or unobserved components models), sample survey methodology, goodness-of-fit tests, and sample distribution functions, with emphasis on applications in the social sciences. He served terms as president of the Royal Statistical Society and the International Statistical Institute.

Geoffrey Stuart Watson (1921–1998) was born in Victoria, Australia, and earned degrees at Melbourne University and North Carolina State University. After a visit to the University of Cambridge, he returned to Australia, working at Melbourne and then the Australian National University. Following periods at Toronto and Johns Hopkins, he settled at Princeton. Throughout his wide-ranging career, he made many notable accomplishments and important contributions, including the Durbin–Watson test for serial correlation, the Nadaraya–Watson estimator in nonparametric regression, and methods for analyzing directional data.

Leslie G. Godfrey (1946–) was born in London and earned degrees at the Universities of Exeter and London. He is now a professor of econometrics at the University of York. His interests center on implementation and interpretation of tests of econometric models, including nonnested models.

Trevor Stanley Breusch (1953–) was born in Queensland and earned degrees at the University of Queensland and Australian National University (ANU). After a post at the University of Southampton, he returned to work at ANU. His background is in econometric methods and his recent interests include political values and social attitudes, earnings and income, and measurement of underground economic activity.

Also see

[R] [regress](#) — Linear regression

[TS] [tsset](#) — Declare data to be time-series data

[R] [regress postestimation](#) — Postestimation tools for regress

[R] [regress postestimation diagnostic plots](#) — Postestimation plots for regress